

Cepat Mahir Visual Basic 6.0

Krisna D. Octovhiana
mail4krisna@yahoo.com

Lisensi Dokumen:

Copyright © 2003 IlmuKomputer.Com

Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari IlmuKomputer.Com.

Bab 9 Penggunaan Array

1.1 Apa itu Array ?

Array merupakan sekumpulan nilai data yang “dikelompokkan” dalam sebuah variabel. Array digunakan bila ada beberapa nilai data yang tipe datanya sama dan akan mendapat perlakuan yang sama pula. Misalnya, ada 10 nilai data dengan tipe string dan akan diolah dengan cara yang sama, maka akan lebih mudah jika menggunakan sebuah array dibandingkan bila menggunakan 10 variabel yang berbeda. Setiap nilai data di dalam sebuah array disebut **elemen array** dan masing-masing dibedakan dengan nomer indeksinya.

1.2 Menggunakan Array

Sebuah array dideklarasikan dengan cara yang sama dengan variabel, yaitu menggunakan perintah Dim, kemudian diikuti dengan tanda kurung dan jumlah elemen arraynya dikurangi 1 :

Dim nama_array(jumlah_elemen - 1) As tipe_data

Contoh :

Dim NamaSiswa(99) As String → array NamaSiswa akan mempunyai elemen sebanyak 100 dengan nomer indeks mulai dari 0 s/d 99.

Selanjutnya untuk mengisi nilai data ke dalam array :

nama_array(no_indeks) = nilai_data

Contoh :

```
NamaSiswa(0) = "umar"  
NamaSiswa(1) = "fatimah"  
...
```

Untuk “mengosongkan” nilai data, bisa menggunakan struktur kontrol For...Next :

```
For i = 0 To 99  
    NamaSiswa(i) = ""  
Next i
```

Lebih mudah bukan ?

Catatan :

- Agar nomer indeks array dimulai dari 1 (bukan 0), bisa menggunakan perintah **Option Base 1** sebelum perintah Dim-nya, contoh :

```
Option Base 1  
Dim NamaSiswa(100) As String
```

- Atau bisa juga menggunakan kata **To** di dalam penentuan jumlah elemen array, contoh :

```
Dim NamaSiswa(1 To 100) As String
```

- Untuk mengetahui berapa jumlah elemen di dalam sebuah array, bisa menggunakan perintah **Ubound(nama_array)**, contoh :

```
Dim NamaSiswa(1 To 100) As String  
Dim JmlElemen As Integer  
  
JmlElemen = Ubound(NamaSiswa) → hasilnya = 100
```

- Bila jumlah elemen array yang dibutuhkan tidak diketahui atau ingin bisa diubah-ubah, maka bagian jumlah_elemen pada perintah Dim tidak perlu diisi, contoh :

```
Dim NamaSiswa() As String
```

Teknik ini biasa disebut sebagai array dinamis, biasanya untuk mengisi data yang berkembang terus, dan belum bisa ditentukan di awal jumlah elemennya.

- Untuk mengubah jumlah elemen array digunakan perintah Redim :

```
Redim [Preserve] nama_array(jumlah_elemen)
```

Contoh :

```
Redim NamaSiswa(1 To 150) → jumlah elemen array NamaSiswa menjadi 150 dengan  
nomer indeks dari 1 s/d 150.
```

Penggunaan kata Preserve dapat dilihat pada contoh program.

1.3 Contoh Program : Array Test


Aktifkan VB 6 melalui tombol Start.

Buka kembali project **Latihan.vbp**.

Tambahkan Form baru ke dalam Project.

Pada Jendela Form buatlah UI seperti ini :

(Gunakan komponen ListBox, ComboBox, TextBox dan CommandButton)



Pengaturan property setiap object-nya adalah sebagai berikut :

Object	Properties	Value
Form8	Caption StartPosition	Array Test 2 – CenterScreen
List1	-	-
Combo1	-	-
Text1	Text	<kosong>
Command1	Caption	Input
Command2	Caption	Redim

Buka Jendela Code dan pada bagian Code Editor ketikkan kode programnya sebagai berikut :

```
Dim NamaSiswa(4) As String

Private Sub Form_Load()
    Dim i As Integer

    For i = 0 To 4
        Combo1.AddItem i
    Next i
    Combo1.ListIndex = 0
End Sub

Private Sub Command1_Click()
    Dim no As Integer, i As Integer

    no = CInt(Combo1.Text)
    NamaSiswa(no) = InputBox("Tuliskan nama siswa no : " & no, "Input
Nama Siswa")
    If NamaSiswa(no) <> "" Then
        List1.Clear
        For i = 0 To 4
            List1.AddItem "NamaSiswa(" & i & ")=" & NamaSiswa(i)
        Next i
    End If
End Sub
```

Simpan Form8 (nama file : **Lat8.frm**).

Catatan : Anda bisa men-download file Lat8.frm dan file latihan-latihan sebelumnya.

Klik menu Project > Project1 Properties lalu klik tab General. Gantilah Startup Object-nya menjadi Form8.

Coba jalankan Project1 :

- Pilih nomer indeks elemen array NamaSiswa melalui Combo1 (antara 0 s/d 4)
- Klik Command1 → akan muncul kotak input untuk mengisi nama siswa
- Ketikkan nilai data untuk elemen array dengan nomer indeks yang dipilih, misalnya Salman

- Klik tombol OK
- Isi setiap elemen array NamaSiswa akan muncul di dalam List1
- Ulangi langkahnya untuk mengisi elemen array nomer indeks lainnya

Penjelasan kode program :

```
Dim NamaSiswa(4) As String
Private Sub Form_Load()
    Dim i As Integer

    For i = 0 To 4
        Combo1.AddItem i
    Next i
    Combo1.ListIndex = 0
End Sub
```

Deklarasi array NamaSiswa dengan jumlah elemen 5
Pada saat form di-load ke dalam memori :

Isi Combo1 dengan angka 0 s/d 4
(sesuai dengan no indeks untuk setiap elemen di dalam
array NamaSiswa)

```
Private Sub Command1_Click()  
    Dim no As Integer, i As Integer
```

```
    no = CInt(Combo1.Text)
```

```
    NamaSiswa(no) = InputBox("Tuliskan  
    nama siswa no : " & no, "Input Nama  
    Siswa")
```

```
    If NamaSiswa(no) <> "" Then
```

```
        List1.Clear
```

```
        For i = 0 To 4
```

```
            List1.AddItem "NamaSiswa(" & i  
                & ")=" & NamaSiswa(i)
```

```
        Next i
```

```
    End If
```

```
End Sub
```

Ketika user meng-klik Command1 :

Simpan nomer indeks yang dipilih ke dalam variabel no

Catatan :

Perintah CInt digunakan untuk mengkonversi sebuah nilai data menjadi bertipe integer.

Tampilkan kotak input untuk mengisi nilai data elemen array dengan nomer indeks yang dipilih.

Jika kotak input tidak kosong maka :

Hapus isi List1.

Tampilkan nilai data setiap elemen array NamaSiswa.

Catatan :

Jika tombol Cancel pada kotak input di-klik sama artinya variabel ataupun array yang menyimpan hasil input bernilai kosong ("").

Modifikasi kode programnya menjadi sebagai berikut :

```
Dim NamaSiswa() As String
```

```
Private Sub Form_Load()
```

```
    Dim i As Integer
```

```
    ReDim NamaSiswa(1 To 5)
```

```
    For i = 1 To 5
```

```
        Combo1.AddItem i
```

```
    Next i
```

```
    Combo1.ListIndex = 0
```

```
End Sub
```

```
Private Sub Command1_Click()
```

```
    Dim no As Integer, i As Integer
```

```
    no = CInt(Combo1.Text)
```

```
    NamaSiswa(no) = InputBox("Tuliskan nama siswa no : " & no, "Input Nama Siswa")
```

```
    If NamaSiswa(no) <> "" Then
```

```
        List1.Clear
```

```
        For i = 1 To UBound(NamaSiswa)
```

```
            List1.AddItem "NamaSiswa(" & i & ")=" & NamaSiswa(i)
```

```
        Next i
```

```
    End If
```

```
End Sub
```

```
Private Sub Command2_Click()  
    Dim num As Integer, i As Integer  
  
    If Not IsNumeric(Text1.Text) Then Exit Sub  
  
    num = CInt(Text1.Text)  
  
    ReDim NamaSiswa(1 To num)  
  
    Combo1.Clear  
    List1.Clear  
    For i = 1 To UBound(NamaSiswa)  
        Combo1.AddItem i  
        List1.AddItem "NamaSiswa(" & i & ")=" & NamaSiswa(i)  
    Next i  
    Combo1.ListIndex = 0  
End Sub
```

Coba jalankan Project1 :

- Pilih nomer indeks elemen array NamaSiswa melalui Combo1 (antara 1 s/d 5)
- Klik Command1 → akan muncul kotak input untuk mengisi nama siswa
- Ketikkan nilai data untuk elemen array dengan nomer indeks yang dipilih, misalnya Salman
- Klik tombol OK
- Isi setiap elemen array NamaSiswa akan muncul di dalam List1
- Ketikkan jumlah elemen array yang diinginkan pada Text2, lalu klik Command2
- Jumlah elemen array NamaSiswa akan berubah

Catatan :

- Setelah perintah Redim maka jumlah array akan diubah, tetapi mengakibatkan isi setiap nilai data di dalam array tersebut dikosongkan.
- Agar perintah Redim tidak mengosongkan isi setiap nilai data di dalam array-nya, gunakan kata Preserve setelah perintah Redim-nya, contoh :

```
ReDim Preserve NamaSiswa(1 To num)
```
